

Dodatak A


Pomoćni programi

A.1 Scilab


Za proračune koji zahtjevaju rad sa matricama većim od 2×2 koristimo program Scilab koji se može naći na www.scilab.org

Instalacija programa je standardna Windows instalacija.

Pri prvom pokretanju programa glavni prozor programa će izgledati kao na slici A.1



Slika A.2: SciNotes prozor je potrebno povući za plavu traku ispod alatne trake i pustiti iznad komandnog prozora da se uklopi u glavni prozor.



Slika A.3: Izgled glavnog prozora programa konfigurisan za ugodan rad. U gornjem dijelu je editor skripti, a u donjem komandni prozor gdje će se ispisati rezultati.

Centralni dio je komandni prozor. Možete probati unijeti 2+3 i pritisnuti tipku **Enter**. Scilab će proračunati vrijednost i ispisati rezultat u komandnom prozoru. Sve Scilab komande se mogu unositi direktno u komandni prozor, međutim u našim proračunima uvijek ćemo imati kompleksne probleme koje želimo imati snimljene u datoteku na disku. Stoga, kliknite na meni **Applications** → **SciNotes**. Otvoriće se editor za editovanje Scilab skripti. Sad još možete zbog bolje preglednosti radnog okruženja, SciNotes editor uklopiti u glavni prozor programa. Kliknite lijevom tasterom miša na svijetlo plavu traku ispod toolbara i držeći pritisnuto dugme miša povucite na vrh komandnog prozora.

Slika A.1

Scilab skripte snimamo na disk standardnim Windows komandama sa menija ili alatne trake.

Listing A.1: Jednostavna skripta za rješavanje sistema jednačina

```

1 clear;
2 clc;
3
4 K = [
5 30 4 2
6 5 45 5
7 3 4 23];
8
9 F = [
10 20
11 30
12 5];
13
14 u = K \ F;
15
16 for i = 1:size(u, 'r')
17     mprintf("\n%15.4f", u(i));
18 end;

```

Alatna traka i traka menija se prilagođavaju aktivnom prozoru. Kad je aktivan SciNotes prozor, na traci menija imamo meni **Execute** gdje su komande za pokretanje izvršavanja skripti. Pošto smo na liniji 17 ispis formatirali sa `mprintf`, skriptu ćemo izvršiti sa **Execute** → ... file with no echo.


Ako hoćemo da Scilab ispiše sve rezultate skriptu ćemo pokrenuti sa **Execute** → ... file with echo.

Brojevi linija lijevo nisu dio Scilab koda već samo za referisanje u ovom tekstu.


Stavlajući znak ; na kraj komande sprečavamo ispis rezultata komande ako postoji. Rezultate ćemo ispisati kontrolisano na liniji 17.

Komanda `clc` briše komandni prozor, a komanda `clear` briše postojeće promjenjive.

Zadatak 1 Za rešetku na silci A.4 odrediti pomjeranja čvorova i presječne sile u štapovima.



$$\begin{aligned}
 L &= 100\text{cm} \\
 a &= 0.7 \\
 b &= 0.8 \\
 E &= 2.1 \cdot 10^5 \text{ N/mm}^2 \\
 A &= 1 \text{ cm}^2
 \end{aligned}$$

Slika A.4**Rješenje****Slika A.5:** Označavanje čvorova, štapova i stepeni slobode kretanja.**Listing A.2:** Matrica krutosti štapa rešetke u globalnom koordinatnom sistemu, `Ke_resetka4x4.sci`

```

19 function Ke = Ke_resetka4x4(x1,y1,x2,y2,E1,A)
20 // Ke_resetka4x4 - Matrica krutosti ravnog resetkastog štapa
21 // Ke_resetka4x4(x1,y1,x2,y2,E1,A)
22
23 Len = sqrt((x2 - x1)^2 + (y2 - y1)^2);
24 ka = E1*A/Len;
25 c = (x2 - x1)/Len;
26 s = (y2 - y1)/Len;
27 Ke = ka*[c*c c*s -c*c -c*s
28         c*s s*s -c*s -s*s
29         -c*c -c*s c*c c*s
30         -c*s -s*s c*s s*s];
31 endfunction

```

Listing A.3: Matrica krutosti štapa rešetke u lokalnom koordinatnom sistemu. Trebaće nam za proračun sila u lokalnom koordinatnom sistemu. Matrica krutosti u prethodnom listingu se dobija rotirajući komponentalne krutosti iz lokalnog u globalni koordinatni sistem. Dakle mogla se implementirati samo ova matrica i matrica transformacije a onda za vrijeme izvršavanja premnožiti $K_g = T^T K_{lok} T$.

```

32 function Ke = Ke_resetka2x2(x1,y1,x2,y2,E1,A)
33 // Ke_resetka2x2 - Matrica krutosti ravnog resetkastog štapa
34 // Ke_resetka2x2(x1,y1,x2,y2,E1,A)
35
36 Len = sqrt((x2 - x1)^2 + (y2 - y1)^2);
37 ka = E1*A/Len;
38 Ke = ka*[
39     1 -1
40    -1 1];
41 endfunction

```

Listing A.4: Matrica transformacije. Trebaće nam za transformaciju vektora pomjeranja iz globalnog u lokalni koordinatni sistem. , Transformacija.sci

```

42 function rezultat = Transformacija(x1,y1,x2,y2)
43 // Transformacija - Transformacija koordinata iz
44 // lokalnog u globalni koordinatni sistem
45 // Transformacija(x1,y1,x2,y2)
46
47 duzina = sqrt((x2 - x1)^2 + (y2 - y1)^2);
48
49 c = (x2 - x1)/duzina;
50 s = (y2 - y1)/duzina;
51
52 rezultat = [
53     c s 0 0
54     0 0 c s];
55 endfunction

```

Konačno možemo napisati skriptu za proračun rešetke.

Listing A.5: Skripta za proračun rešetke

```

56 // Statika konstrukcija 1 ..... //
57 // Zadatak 1 .....//
58
59 // Funkcija koja formira matricu krutosti je u
60 // fajlu Ke_resetka4x4.sce
61 // duzina cm; sila N;
62
63 // komanda koja briše sve promjenjive trenutno raspoložive.
64 // promjenjive se mogu vidjeti prozoru "Variable Browser"
65 clear;
66
67 // komanda cisti komandnu konzolu
68 clc;
69
70 // učitavanje funkcije Ke_resetka4x4 koja se nalazi u
71 // istom folderu u dokumentu Ke_resetka4x4.sci
72 exec('./Ke_resetka4x4.sci', -1);
73 exec('./Ke_resetka2x2.sci', -1);
74 exec('./Transformacija.sci', -1);
75
76 // ako na kraju izraza ne stavimo znak ; onda ce rezultat
77 // izvršenja komande biti printan na komandnoj konzoli.
78 // Ovdje cemo uglavnom koristiti znak ; a rezultate cemo

```

```

79 // formatirati sa funkcijom mprintf
80 // Program cemo pokretati sa
81 // Execute -> ... file with no echo
82 // tako da ce se printati samo ono sto
83 // definisemo sa mprintf
84
85 //konstante
86 a = 0.7;
87 b = 0.8;
88 l = 100.0;
89
90 A = 1;
91 E = 2.1*10^7;
92
93 // matrica se u Scilabu moze definisati u jednom redu
94 //koristeci ; znak za ogranicavanje redova matrice
95 // A = [a11 a12; a21 a22]
96 // (primjetiti ; znak nakon prvog reda)
97
98 // tacka zarez nakon zatvaranja uglaste zagrade ponovo
99 // znaci ne ispisivanje rezultata komande
100
101 // matrica se takodje moze definisati bez znaka ; ali onda
102 // svaki pojedinačni red moramo zapisati u novom redu
103 // A = [a11 a12
104 // a21 a22]
105
106 // koordinate cvorova = koordinate(brojCvora, 1 = X[2 = Y])
107 koordinate = l*[0 b
108 0 0
109 2 b
110 2 0
111 2+2*a b
112 2+2*a 0
113 4+2*a b
114 4+2*a 0]
115
116 // broj cvora na stapu = stap(brStapa, 1 = cvori[2 = cvorj])
117 stap = [1 2
118 1 3
119 1 4
120 2 4
121 3 4
122 3 5
123 4 5
124 4 6
125 5 6
126 5 7
127 5 8
128 6 8
129 7 8];
130
131 // brojSSK = SSK[brojCvora, 1 = ssk X [2 = ssk Y][-1 =
132 // nepokretno ]
133 // (broj jednacine ili adresa u KGlob)
134
135 SSK = [1 2
136 3 -1
137 4 5
138 6 7
139 8 9
140 10 11
141 12 13
142 -1 -1];
143
144 // Funcija zeros(ni, nj) formira nula matricu
145 // sa ni redova i nj kolona
146 KGlob = zeros(13,13);
147
148 // for petlja u scilabu ima oblik
149 //
150 // for i = 1:n
151 //
152 // end
153 // gdje je i brojac a n broj koraka
154 //
155 // funkcija size(A, 'r'['c']) vraća
156 // broj redova matrice A koristeci selektor 'r'
157 // broj kolona matrice A koristeci selektor 'c'
158 //
159 //

```

```

160 for i = 1:size(stap, 'r')
161
162     ni = stap(i, 1);
163     nj = stap(i, 2);
164
165     x1 = koordinate(ni, 1);
166     y1 = koordinate(ni, 2);
167
168     x2 = koordinate(nj, 1);
169     y2 = koordinate(nj, 2);
170
171     Ke = Ke_resetka4x4(x1, y1, x2, y2, E, A);
172
173     addr(1) = SSK(ni, 1);
174     addr(2) = SSK(ni, 2);
175     addr(3) = SSK(nj, 1);
176     addr(4) = SSK(nj, 2);
177
178     // znaci poredjenja u logickim izrazima
179     // == jednako
180     // ~= nije jednako
181     // > vece
182     // >= vece ili jednako
183     // < manje
184     // <= manje ili jednako
185
186     // logicki operatori u logickim izrazima
187     // & i
188     // | ili
189
190     for j = 1:4
191         for k = 1:4
192             if ( addr(j) ~= -1) & (addr(k) ~= -1) )
193                 KGlob( addr(j), addr(k) ) = ..
194                     KGlob( addr(j), addr(k) ) + Ke(j, k);
195             end
196         end // for k
197     end // for j
198
199 end //for i za sve stapove
200
201 // formiranje globalnog vektora opterecenja
202 RGlob = zeros(13,13);
203 RGlob(9,1) = -100;
204
205 //rjesavanje sistema jednacina
206 UGlob = KGlob \ RGlob;
207
208 // uzimanje pomjeranja cvora iz u -> cvorPomak
209 pomakCvora = zeros(8, 2);
210
211 for i = 1:8
212     addr(1) = SSK(i, 1);
213     addr(2) = SSK(i, 2);
214
215     if addr(1) ~= -1
216         pomakCvora (i, 1) = UGlob( addr(1) );
217     end
218     if addr(2) ~= -1
219         pomakCvora (i, 2) = UGlob( addr(2) );
220     end
221 end
222
223 mprintf("\n===== REZULTATI PRORACUNA\n");
224 mprintf("\n===== POMJERANJA CVOROVA\n");
225 mprintf("%4s%15s%15s", "Cvor", "ux", "uy");
226 mprintf("\n=====");
227 for i = 1:size(pomakCvora, 'r')
228     mprintf("\n%4d", i);
229     mprintf("%15.6f", pomakCvora(i, 1));
230     mprintf("%15.6f", pomakCvora(i, 2));
231 end
232
233 mprintf("\n\n===== SILE PO STAPOVIMA\n");
234 for i = 1:size(stap, 'r')
235
236     ni = stap(i, 1);
237     nj = stap(i, 2);
238
239     x1 = koordinate(ni, 1);
240     y1 = koordinate(ni, 2);
241

```

```

242 x2 = koordinate(nj, 1);
243 y2 = koordinate(nj, 2);
244
245 addr(1) = SSK(ni, 1);
246 addr(2) = SSK(ni, 2);
247 addr(3) = SSK(nj, 1);
248 addr(4) = SSK(nj, 2);
249
250 for j = 1:4
251     if addr(j) ~= -1 then
252         uglob(j) = UGlob(addr(j))
253     else
254         uglob(j) = 0.0
255     end
256 end
257
258 T = Transformacija(x1, y1, x2, y2);
259
260 // vektor pomjeranja cvorova u lokalnom
261 // koordinatnom sistemu stapa
262 u = T * uglob
263
264 // matrica krutosti
265 Ke = Ke_resetka2x2(x1, y1, x2, y2, E, A);
266
267 // proračun vektora sila na stapu
268 f = Ke * u;
269
270 mprintf("\n----- %4s%3d", "STAP", i);
271 mprintf("\n%11.2f", f(1:2,1));
272
273 end;

```

Listing A.6: Rezultat proračuna

```

274 ===== REZULTATI PRORACUNA
275
276 ===== POMJERANJA CVOROVA
277 Cvor      ux      uy
278 =====
279 1      -0.000567  -0.000141
280 2      -0.002549   0.000000
281 3      -0.001449  -0.007849
282 4      -0.002549  -0.007849
283 5      -0.002066  -0.009848
284 6      -0.001499  -0.009848
285 7      -0.002066   0.000000
286 8       0.000000   0.000000
287
288 ===== SILE PO STAPOVIMA
289
290 ----- STAP 1
291      37.04
292     -37.04
293 ----- STAP 2
294      92.59
295     -92.59
296 ----- STAP 3
297     -99.73
298      99.73
299 ----- STAP 4
300       0.00
301       0.00
302 ----- STAP 5
303       0.00
304       0.00
305 ----- STAP 6
306      92.59
307     -92.59
308 ----- STAP 7
309      74.65
310     -74.65
311 ----- STAP 8
312     -157.41
313      157.41
314 ----- STAP 9
315       0.00
316       0.00
317 ----- STAP 10
318       0.00
319       0.00
320 ----- STAP 11

```

```

321      169.53
322     -169.53
323 ----- STAP 12
324     -157.41
325      157.41
326 ----- STAP 13
327       0.00
328       0.00

```